

Math 2280-2

§5.5 Matrix exponentials

use the LinearAlgebra package (instead of linalg)

```
> with(LinearAlgebra):
```

Define the matrix we work with (row by row)

```
> A:=Matrix(3,3,[3,4,5, 0,5,4, 0,0,3]);
```

$$A := \begin{bmatrix} 3 & 4 & 5 \\ 0 & 5 & 4 \\ 0 & 0 & 3 \end{bmatrix} \quad (1)$$

The LinearAlgebra package has a function to compute the eigenvalues and eigenvectors of some matrix.

If the matrix is defective, some of the eigenvalues will be accompanied by an "eigenvector" that is all zeros.

This does not make sense, as eigenvectors are non-zero

```
> Eigenvectors(A);
```

$$\begin{bmatrix} 5 \\ 3 \\ 3 \end{bmatrix}, \begin{bmatrix} 2 & 1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (2)$$

The first output is a vector whose entries are the eigenvalues of A. The i-th column of the second output (a matrix) is the eigenvector associated with the i-th eigenvalue of the vector of eigenvalues.

```
> Id:=IdentityMatrix(3);
```

$$Id := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (3)$$

When the matrix is defective we no longer look for a basis of eigenvectors, but one of generalized eigenvectors. Since $\lambda=5$ is simple, the associated eigenvector is also a generalized eigenvector. For $\lambda=3$ (multiple) the generalized eigenvectors belong to the space

```
> NullSpace((A-3*Id)^2); # generalized eigenspace for lambda = 3
```

$$\left\{ \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ -2 \\ 1 \end{bmatrix} \right\} \quad (4)$$

We have squared the matrix $(A-3*Id)$ because $\lambda=3$ is an eigenvalue of algebraic multiplicity 2.

We now have a basis of generalized eigenvectors of A. We use Theorem 5.5.3 to construct linearly independent solutions...

```
> sol1:=t->exp(5*t)*Vector([2,1,0]);
sol2:=t->exp(3*t)*Vector([1,0,0]);
sol3:=t->simplify(exp(3*t)*(Id + t*(A-3*Id)).Vector([0,-2,1]));
sol1 := t -> e^{5t} Vector([2, 1, 0])
```

$$\begin{aligned} \text{sol2} &:= t \rightarrow e^{3t} \text{Vector}([1, 0, 0]) \\ \text{sol3} &:= t \rightarrow \text{simplify}((e^{3t} (\text{Id} + t(A - 3 \text{Id}))).\text{Vector}([0, -2, 1])) \end{aligned} \quad (5)$$

> **sol3(t);**

$$\begin{bmatrix} -3 e^{3t} t \\ -2 e^{3t} \\ e^{3t} \end{bmatrix} \quad (6)$$

... which serve to define a fundamental matrix solution...

> **Phi := t -> Matrix(3, 3, [sol1(t), sol2(t), sol3(t)]);**
Phi(t);

$$\Phi := t \rightarrow \text{Matrix}(3, 3, [\text{sol1}(t), \text{sol2}(t), \text{sol3}(t)])$$

$$\begin{bmatrix} 2 e^{5t} & e^{3t} & -3 e^{3t} t \\ e^{5t} & 0 & -2 e^{3t} \\ 0 & 0 & e^{3t} \end{bmatrix} \quad (7)$$

> **Phi(0);**

$$\begin{bmatrix} 2 & 1 & 0 \\ 1 & 0 & -2 \\ 0 & 0 & 1 \end{bmatrix} \quad (8)$$

... and the exponential (by eq. (36) in the book)

> **eAt := Phi(t) . MatrixInverse(Phi(0));**

$$eAt := \begin{bmatrix} e^{3t} & 2 e^{5t} - 2 e^{3t} & 4 e^{5t} - 4 e^{3t} - 3 e^{3t} t \\ 0 & e^{5t} & 2 e^{5t} - 2 e^{3t} \\ 0 & 0 & e^{3t} \end{bmatrix} \quad (9)$$

Compare to the result obtained via the builtin Maple command:

> **MatrixExponential(A*t);**

$$\begin{bmatrix} e^{3t} & 2 e^{5t} - 2 e^{3t} & 4 e^{5t} - 4 e^{3t} - 3 e^{3t} t \\ 0 & e^{5t} & 2 e^{5t} - 2 e^{3t} \\ 0 & 0 & e^{3t} \end{bmatrix} \quad (10)$$

Exercise 5.2.8,

we had $x' = A x$ with

> **A := Matrix(2, 2, [1, -5, 1, -1]);**

$$A := \begin{bmatrix} 1 & -5 \\ 1 & -1 \end{bmatrix} \quad (11)$$

The following is a general solution to $x' = Ax$

> **MatrixExponential(A*t) . Vector([a, b]);**

$$\begin{bmatrix} \left(\cos(2t) + \frac{1}{2} \sin(2t) \right) a - \frac{5}{2} \sin(2t) b \\ \frac{1}{2} \sin(2t) a + \left(\cos(2t) - \frac{1}{2} \sin(2t) \right) b \end{bmatrix} \quad (12)$$

Other examples from the notes

> **B:= Matrix(2,2,[0,1,-1,0]);**

$$B := \begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix} \quad (13)$$

> **MatrixExponential(B*t);**

$$\begin{bmatrix} \cos(t) & \sin(t) \\ -\sin(t) & \cos(t) \end{bmatrix} \quad (14)$$

A Nilpotent matrix

> **N:=Matrix(3,3,[0,1,1,0,0,2,0,0,0]);**

$$N := \begin{bmatrix} 0 & 1 & 1 \\ 0 & 0 & 2 \\ 0 & 0 & 0 \end{bmatrix} \quad (15)$$

> **N^2; N^3;**

$$\begin{bmatrix} 0 & 0 & 2 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (16)$$

> **MatrixExponential(N*t);**

$$\begin{bmatrix} 1 & t & t^2 + t \\ 0 & 1 & 2t \\ 0 & 0 & 1 \end{bmatrix} \quad (17)$$

A nilpotent matrix plus a multiple of the identity.

> **A:= 3*Id+N;**

$$A := 3 \text{ Id} + \begin{bmatrix} 0 & 1 & 1 \\ 0 & 0 & 2 \\ 0 & 0 & 0 \end{bmatrix} \quad (18)$$

> **MatrixExponential(A*t);**

(19)



$$\text{MatrixExponential} \left(\left(3 \text{Id} + \begin{bmatrix} 0 & 1 & 1 \\ 0 & 0 & 2 \\ 0 & 0 & 0 \end{bmatrix} \right) t \right)$$

(19)